

Contents	Page
1. MACHINE DESCRIPTION	1
2. REQUIREMENTS	1
3. COMPONENTS SELECTED	1
4. IMPLEMENTATION	2
5. APPENDIX: DERIVATION OF ECAM PROFILE	4

1. Machine Description

Web cut machines are used where a material such as fabric is fed continuously at high speed and needs to be cut to a specific length. The material is cut on the fly without stopping, allowing higher throughput to be achieved than could be realized if the material was stopped every time a cut is made. In the application described here, a carriage carrying the cutting tool must be accelerated to meet the speed of the feed material before the cut occurs. The solution to this type of application is to use an electronic cam to synchronize the motion of the shear with the motion of the conveyor.

In a flying shear web-cut application, the material to be cut is fed on a continuous conveyor that is driven by an open-loop motor (*figure 1*). The shear is mounted on a carriage under servo control that runs parallel to the conveyor. The shear accelerates to meet the velocity of the material to perform the cut at the correct location. When the cut is complete, the shear rapidly decelerates and moves back to the starting position to begin the next cutting cycle. This results in equal length pieces of material being fed to the next machine process.

There are three major electro-mechanical systems which make up this flying shear application:

- (1) **In-feed conveyor (master)** – This is run by an open-loop motor, which is not commanded by the motion controller. An encoder is attached to the motor, which feeds position information to the controller.
- (2) **Flying shear mechanism (slave)** – The flying shear axis is driven by a closed-loop servo motor, which is commanded by the motion controller. The position of this slave is determined by the position of the master as well as the defined cam profile. This servo motor is




Figure 1.

connected to a lead screw, which drives the knife to match the speed of the conveyor.

(3) **Knife** – This application uses an output bit to fire a pneumatically driven knife downward into the fabric at the proper position. Other applications may use rotating knives or blades cutting across the material.

2. Requirements

This section summarizes the requirements for the machine described above:

- (1) *One axis to move the shear*
- (2) *Forward and reverse limit switches for the shear axis*
- (3) *An additional encoder input for the master encoder*
- (4) *Electronic Cam to simulate the motion of a mechanical cam*
- (5) *Consistent material cut length. A registration eye is not required*
- (6) *A digital output bit to activate the knife*
- (7) *Stand-alone operation (no operator interface or host computer)*

3. Components Selected

This section describes the Galil hardware and software products chosen to implement the machine's control system. Below is a complete bill of materials followed by a description of major components.

Table 1. Bill of Materials for Flying Shear Control System

Part Number	Description	Unit Price (U.S.) Qty 1 / qty 100
DMC-1416-BRUSHLESS	Stand-Alone, Single Axis-Motion Controller with Brushless Motor Amplifier	\$1095/\$895
BLM-N23-50-1000 or equivalent	NEMA 23 Brushless Servo Motor with 1000-line Encoder	Consult mfg.
CPS-6-48 or equivalent	Power Supply 48V, 6A	Consult mfg.
ICM-1460	Interconnect Module, Provides Screw Terminals for Access to all Controller Signals	\$145/\$95
CABLE-37-pin D	37-pin cable for use with ICM-1460 Above	\$25/\$15
CABLE-9PIN-D	2 Meter, Serial Communication Cable	\$10
WSDK Servo Tuning Software	Servo Tuning and Analysis Software	\$195 (one time)
ECAM32	Electronic Cam Setup Utility (allows unevenly-spaced points)	\$195 (one time)

Controller: DMC-1416

Since the flying shear application does not require a host computer to be connected to the controller, we choose a stand-alone controller. The DMC-1416 controller supports both Ethernet and RS-232 communications as well as non-volatile program memory, making it ideal for stand-alone applications.

The DMC-1416 is also packaged with an integrated PWM amplifier. The DMC-1416-BRUSHLESS is capable of driving a brushless motor at up to 6 Amps continuous at 60 volts. The DMC-1416 can handle the most demanding applications with such features as an extra encoder input for electronic cam and uncommitted I/O.



Figure 2. DMC-1416

Motor: BLM-N23-50-1000

For maintenance-free operation, we choose a brushless motor. Galil’s NEMA 23 #BLM-N23-50-1000 brushless motor, or equivalent, is appropriate because the axis requires less than 0.3 Nm of continuous torque. An incremental encoder with 1000 cycles per revolution is installed on the motor resulting in 4000 quadrature counts per revolution. Hall sensors are not required on the motor as the incremental encoder provides commutation tracks for input to the amplifier.

4. Implementation

Electronic Cam

The key to this application is the use of the ECAM (electronic cam) feature of the Galil controller. With ECAM, any slave axis or set of slave axes can be linked to a master axis to simulate the motion of a mechanical cam. This enables periodic synchronization of one or more axes of motion to one master. The master axis can be any motor-driven axis or encoder.

Galil motion controllers treat the ECAM function as a table-based relationship of slave positions versus master positions over one cycle. The flying shear application defines one cycle as the distance the master encoder moves for one complete cycle of the slave axis (the cut length). The slave axis cycle consists of a rapid acceleration to match the conveyor speed, slewing at speed during the cut, rapid deceleration, and finally a return to the start position.

Advantages of an electronic cam over a mechanical cam include the ability to programmatically change the cut length and the profile. Also, there’s no need to change expensive mechanical parts like mechanical cams to change the profile.

Program Structure

It is assumed that the entire cam cycle has been defined at this point. Please refer to the appendix for a detailed derivation of the cam table. The final step in the design process is to put all of these pieces together into a program for the motion controller. The pseudo code for the program is as follows:

- (1) **Program Start** – #AUTO (runs automatically on power up).
- (2) **Home Axes** – Send the flying shear axis to its reverse limit switch and then an index pulse. This insures the shear carriage is properly aligned on the lead screw before beginning a cycle.

(3) Define Parameters:

(a) Define X-Axis auxiliary encoder as ECAM master and X-Axis main encoder as ECAM slave (Galil EA command).

(b) Define slave and master modulus for a single cycle (Galil EM command). This would be 0 mm for the slave (because it advances and returns to 0 within each cycle) and the length of the cut for the master.

(c) Define master interval length (Galil EP command).

(4) Enter ECAM table in encoder counts. –This is based on the distances calculated in the appendix.

Table 2. Master position versus slave position (see Appendix for derivation)

Interval	Master Position (encoder counts)	Slave Position (encoder counts)
1	250	1000
2	500	3000
3	750	5000
4	1000	7000
5	1250	8000
6	1500	7000
7	1750	5000
8	2000	3000
9	2250	1000
10	2500	0

(5) Enable ECAM – Start at 0 position of master aligned with 0 position of slave (Galil commands EB and EG). The ECAM mode must be enabled and the start point for the slave with respect to the master must be indicated.

(6) Begin Slave position loop routine – The slave loop routine is used to fire the cutter once per interval for even material cuts.

(a) If Slave position > Start of cutting zone (MF command) Enable flying shear cutter (SB command)

(b) If Slave position > End of cutting zone (MF command) Disable flying shear cutter (CB command)

(c) Wait for slave to return to 0 (MR command)

(7) Return to Slave position loop beginning

(8) End Program

Program Listing

The complete program used for this article is included here:

```
'This program uses ecam to operate a flying shear
'X axis auxiliary encoder is the master
'X axis main encoder is the slave

#AUTO
'X-axis slave is homed to the reverse limit
switch and then the index pulse

JGX = -10000 ;'jog towards reverse limit switch
BGX          ;'begin motion towards limit
AMX          ;'wait until we hit the limit

JGX = 500    ;'move slowly towards the index
pulse
FIX          ;'find index
BGX          ;'begin motion towards index
AMX          ;'wait until we hit the index.
Position is set to 0.

EAX          ;'Define X-Axis aux encoder as
master and X-Axis main encoder as
slave
EM 0,2500    ;'Define slave and master modulus
for a single cycle
EP 250,0     ;'Define master interval length
ET[0] = 0    ;'Enter ECAM table in encoder
counts
ET[1] = 1000
ET[2] = 3000
ET[3] = 5000
ET[4] = 7000
ET[5] = 8000
ET[6] = 7000
ET[7] = 5000
ET[8] = 3000
ET[9] = 1000
ET[10] = 0

'Enable ecam mode:
EB1          ;'enable master
EG0          ;'engage slave

'Loop to fire flying shear at proper slave
positions:
#CUT
MF 1000      ;'If Slave position > Start of
cutting zone
SB 2         ;'Enable flying shear cutter
MF 7000      ;'If Slave position > End of
cutting zone
CB 2         ;'Disable flying shear cutter
MR 100       ;'Wait for slave to return to 0
JP#CUT       ;'Return to Slave position loop
beginning
EN           ;'End Program
```

5. Appendix: Derivation of ECAM Profile

This section details the derivation of the ECAM table used for the flying shear.

Specifications

The first step is to define the performance specifications for the application. The following numbers define the flying shear application:

cut length: 250 mm
linear speed of material: 500 mm/s
minimum duration of cut: 100 ms

To convert from millimeters to counts, we need to know the following:

Conveyor roll diameter: 12.73 cm Conveyor
encoder resolution: 4000 counts/revolution

$(\pi * 12.73 \text{ cm/rev}) * (1 \text{ rev}/4000 \text{ counts}) * (10 \text{ mm/cm}) = 0.1 \text{ mm/count or } 10 \text{ counts/mm}$

Flying shear encoder resolution: 2000 counts/revolution
Flying shear lead screw pitch: 2.5 cm/revolution

$(2.5 \text{ cm/rev}) * (1 \text{ rev}/2000 \text{ counts}) * (10 \text{ mm/cm}) = 0.0125 \text{ mm/count or } 80 \text{ counts/mm}$

Calculation

Slave Velocity vs. Time Graph

To begin the application design, it is helpful to calculate a slave profile by graphing its velocity as a function of time (*figure 3*). Values needed for this graph may be calculated as follows.

- (1) A cut length of 250 mm at 500 mm/s gives a single cycle time of 500 ms.
- (2) The speed of the slave during cutting must be equal to the speed of the fabric. This is given as 500 mm/s.
- (3) The minimum duration of the actual cut (knife in contact with material) is 100 ms. This is the minimum time during which the slave speed must equal the fabric speed.

Figure 3 shows the slave accelerate to 500 mm/s and cruise at this speed for a duration of 150 ms. This 150 ms is greater than the specified minimum duration of 100 ms. A symmetrical profile was chosen for ease of programming, but a return move with a higher acceleration and speed could increase machine throughput.

Looking at the above graph, the question arises how this graph changes based on master velocity, and will this limit the machine to only operating at one speed? The flying

shear application is designed around a nominal velocity of the master and slave. This means that, while we calculate the positional values based on the nominal velocity, these same values will work at all system speeds (below the critical speed which has a 100 ms cruise region) without recalculating profiles. This is due to the positional relationship defined in the electronic cam. For example, if the master were operating at $\frac{1}{2}$ maximum speed (60 cycles/minute), the single cycle time is 1000 ms, or 1 second and the slave would therefore be operating at 250 mm/s during the cut. The values in this case are scaled based on the speed, but the overall position relationship between master and slave remains constant.




Figure 3. Slave Velocity vs. Time

Master Interval

The next step of the flying shear design is to correlate the master positions with the slave profile positions. The relationship between the two is defined in a tabular format by specifying the slave position at each master interval. A typical ECAM application will have a large number of these intervals, but for simplicity, we will use only 10. Looking at the *Figure 3* Velocity/Time graph of the slave, we can see 10 distinct 50 ms divisions in time, so the master could therefore be divided up into equal position intervals as follows:

Conveyor Speed = 500 mm/s

Single Cycle Time = 500 ms

Conveyor travel in one cycle = 250 mm

Conveyor interval = 25 mm/interval

Slave Positions

We now need to calculate the distance the slave moves in each interval of the master. The slave positions may be calculated using the Velocity/Time graph and standard kinematics equations as follows:

Acceleration/Deceleration Distances:

$$S = \frac{1}{2} * A * t^2$$

The initial acceleration of the slave shown in *Figure 3* indicates an acceleration to 500 mm/s within 50 ms. Using the above equation, we calculate:

$$S = \frac{1}{2} * (500 \text{ mm/s} / 0.05 \text{ s}) * (0.05 \text{ s})^2 = 12.5 \text{ mm}$$

Since we've chosen a symmetrical profile, this distance may be used for all accelerations and decelerations of the slave.

Constant Velocity Distances:

$$S = V * t$$

The constant velocity portion of the slave profile is also shown in *Figure 3*. The speed of 500 mm/s is held for 150 ms on both the positive and negative moves. In order to divide this into 10 intervals, we select a single 50 ms window for our calculation.

$$S = 500 \text{ mm/s} * 0.05 \text{ s} = 25 \text{ mm}$$

This gives a total travel within the constant velocity section of 75 mm.

We can then fill in a table which correlates the master intervals with these slave intervals, and finally the cumulative positions in counts are calculated for the ET command (*Table 3 and Figure 4*).




Figure 4. The cam table

Table 3. Slave position versus master position (the cam table) in relative distance, cumulative distance, and cumulative counts. 10 counts/mm (master), 80 counts/mm (slave)

Interval	Master Position	Slave Position	Cumulative Master Position	Cumulative Slave Position	Cumulative Master Position	Cumulative Slave Position
1	25	12.5	25	12.5	250	1000
2	25	25	50	37.5	500	3000
3	25	25	75	62.5	750	5000
4	25	25	100	87.5	1000	7000
5	25	12.5	125	100	1250	8000
6	25	-12.5	150	87.5	1500	7000
7	25	-25	175	62.5	1750	5000
8	25	-25	200	37.5	2000	3000
9	25	-25	225	12.5	2250	1000
10	25	-12.5	250	0	2500	0

Note the negative slave position intervals. These reflect the return move of the flying shear and insure that the move always starts and ends at the zero point.